
 

Laborator 5:  
CURSORI (partea a I-a) 

 
 

Atunci când se execută o comandă SQL, Oracle Server deschide o zonă de memorie 

(context area) în care comanda este executată. Cursorul este un pointer către această zonă  

În PL/SQL se utilizează două tipuri de cursoare: 

1. implicit: declarat pentru toate instrucţiunile PL/SQL de tip LMD 

(INSERT/UPDATE/DELETE); 

2. explicit: declarat şi gestionat de programator. 

 

 

 

 

1. CURSORUL IMPLICIT 
 

Este declarat de PL/SQL implicit pentru toate comenzile de manipulare a datelor (INSERT, 

UPDATE, DELETE); 

Dacă o instrucţiune LMD nu afectează nici o linie a tabelei nu se generează eroare, însă 

excepţia trebuie tratată folosind atributele speciale ale cursorilor; 

Atributele cursorului implicit: 

 SQL%ROWCOUNT 

 SQL%FOUND 

 SQL%NOTFOUND 

 

Exemplu 

 

Scrieti un program in PL/SQL pentru sterge din copia tabelei EMP angajatul cu ID-ul 7788 si 

a afisa numarul de randuri sterse, folosind atributul %ROWCOUNT. 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

DECLARE 

    v_rez NUMBER(2); 

BEGIN 

    DELETE FROM copy_emp 

    WHERE empno=7788; 

 

    v_rez:=SQL%ROWCOUNT; 

    DBMS_OUTPUT.PUT_LINE (v_rez || ' randuri sterse'); 

END; 



Scrieti un program in PL/SQL pentru a arata utilizarea unui cursor implict fara a folosi niciun 

atribut.  

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

DECLARE 

    emp_ename VARCHAR2(35); 

    emp_sal  VARCHAR2(35); 

 zemp_id emp.empno%type:=234; 

BEGIN 

    SELECT ename, sal 

    INTO   emp_ename, emp_sal 

    FROM   emp 

    WHERE  empno = zemp_id; 

    dbms_output.Put_line ('Employee name: '|| emp_ename||' '||emp_sal); 

    EXCEPTION 

        WHEN no_data_found THEN 

        dbms_output.Put_line ('There is no employee with the ID 

'||to_char(zemp_id)); 

END; 



Scrieti un program in PL/SQL pentru a crea o inregistrare din tabela cu ajutorul atributului  

%ROWTYPE. 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

DECLARE 

    v_empno emp%ROWTYPE; 

 z_emp_no emp.empno%TYPE; 

BEGIN 

    z_emp_no := 7782; 

    SELECT * 

    INTO   v_empno 

    FROM   emp 

    WHERE  empno = z_emp_no; 

 

    dbms_output.Put_line (v_empno.ename||' ' ||v_empno.sal  ||' has an ID of 

'||z_emp_no); 

    EXCEPTION 

        WHEN no_data_found THEN 

            Raise_application_error(-20001, 'The Employee '|| 'is not in the 

database'); 

END; 



2. CURSORUL EXPLICIT  
  

 se foloseşte pentru a procesa individual fiecare linie (înregistrare) returnată de o 

instrucţiune SELECT ce returnează mai multe înregistrări. 

 mulţimea înregistrărilor returnate de o instructiune SELECT este numită mulţime 

rezultat.  

 cursorul păstrează un pointer către linia curentă în cadrul unei mulţimi rezultat. 

 

Verificarea stării unui cursor explicit se realizează prin intermediul următoarelor atribute:  

1. nume_cursor%ISOPEN - evaluat la TRUE în cazul în care cursorul este deschis; 

2. nume_cursor %NOTFOUND - evaluat la TRUE în cazul în care cel mai recent 

FETCH nu a returnat nici o linie; 

3. nume_cursor %FOUND - complementul lui %NOTFOUND; 

4. nume_cursor %ROWCOUNT - are ca valoare numărul liniilor returnate până în 

momentul curent. 

 

Prelucrarea cursorului explicit presupune parcurgerea următoarelor etape: 

1) se declară variabilele în care vor fi încărcate valorile corespunzătoare unei linii din 

cursor; 

 

2) se declară cursorul explicit, specificându-se un nume pentru acesta şi definindu-se 

interogarea de procesat în cadrul lui: 

DECLARE nume_cursor IS SELECT........................; 

 

3) se deschide cursorul prin intermediul instrucţiunii OPEN, care execută interogarea şi 

legarea tuturor variabilelor referite. Înregistrările returnate de interogare sunt desemnate drept 

set activ de date, care pot fi de acum încărcate. 

OPEN nume_cursor; 

 

4) utilizându-se instrucţiunea FETCH, se încarcă linia curentă din cursor în variabile. 

Fiecare încărcare determină mutarea pointerului cursorului la linia următoare din setul activ 

de date. 

FETCH nume_cursor INTO var1, var2,..............; 

 

5) este închis cursorul prin instructiunea CLOSE, care dezafectează setul activ de linii. 

Cursorul poate fi din nou deschis pentru a stabili un nou set activ de linii. 

CLOSE nume_cursor; 

 

 Pentru a procesa liniile unui cursor explicit se defineşte de obicei o buclă pentru 

executarea unui FETCH în fiecare iteraţie. În final, toate liniile din setul activ sunt procesate 

şi un FETCH executat fără succes poziţionează atributul %NOTFOUND pe TRUE.  

 

 Înaintea primului FETCH, %NOTFOUND se evaluează la NULL, ca şi în cazul în 

care FETCH nu se execută niciodată cu succes. 

 

 

 

 

 

 



Exemple: 

 

Scrieti un program in PL/SQL pentru a afişa lista cu numele şi salariul angajaţilor din 

departamentul 30 folosind un cursor explicit. 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

DECLARE 

    cursor emp_cursor is select empno, ename, sal from emp where deptno = 30; 

    v_empno emp.empno%type; 

    v_ename emp.ename%type; 

    v_sal emp.sal%type; 

BEGIN 

    dbms_output.put_line('Lista cu salariariile angajatilor din departamentul 

30'); 

    open emp_cursor; 

    loop 

        fetch emp_cursor into v_empno, v_ename, v_sal; 

        exit when emp_cursor%notfound; 

        dbms_output.put_line('Salariatul '||v_ename||' are salariul: '||v_sal); 

    end loop; 

    close emp_cursor; 

END; 



Pentru o flexibilitate mai mare se poate utiliza o variabilă de tip record pentru 

încărcarea valorilor din cursor. Această variabilă de tip record poate avea aceleaşi atribute ca 

şi cursorul prin specificarea proprietăţii %ROWTYPE.  

În acest caz încărcarea din cursor se va face direct prin instrucţiunea fech var_cursor 

into var_record. 

Exemplul de mai sus poate fi rescris astfel: 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

DECLARE 

    cursor emp_cursor is select empno, ename, sal from emp where deptno = 30; 

--tipul record pt incarcarea valorilor cursorului 

    emp_rec emp_cursor%rowtype; 

BEGIN 

    dbms_output.put_line('Lista cu salariariile angajatilor din departamentul 

30'); 

    open emp_cursor; 

    loop 

        fetch emp_cursor into emp_rec; 

        exit when emp_cursor%notfound; 

        dbms_output.put_line('Salariatul '||emp_rec.ename||' are salariul: 

'||emp_rec.sal); 

    end loop; 

    close emp_cursor; 

END; 



Scrieti un program in PL/SQL pentru a se încarca în tabela MESAJE primii 5 angajaţi (empno 

şi ename) 

 

Pasul 1 – crearea unei noi tabele numite MESAJE: 

 

 

 

 

 

 
 

Pasul 2 – crearea unui cursor pentru a prelua din tabela EMP primele 5 linii: 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

CREATE TABLE mesaje ( cod varchar2(7), nume varchar2(20) ); 

DECLARE  

v_id angajati.id_angajat%type; 

v_nume angajati.nume%type;  

CURSOR c1 IS SELECT id_angajat, nume FROM angajati; 

BEGIN 

OPEN c1; 

FOR i IN 1..5 LOOP 

FETCH c1 INTO v_id, v_nume; 

INSERT INTO mesaje VALUES(v_id, v_nume); 

END LOOP; 

CLOSE c1; 

END; 



 
 

 

Pasul 3 – afisarea continutului noii tabele MESAJE: 

 

 

 

 

 
 

 

 

 

SELECT * FROM mesaje; 



Testul de ieşire din buclă în acest caz se poate face şi cu ajutorul atributului 

%ROWCOUNT: 

Delete from mesaje; 

 

Pasul 1 – Stergerea datelor existente in tabela MESAJE  

 

 

 

 

 
 

Pasul 2 – crearea unui cursor pentru a prelua din tabela EMP primele 5 linii, folosind atributul 

%ROWCOUNT: 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

DELETE FROM mesaje; 

DECLARE  

    v_empno emp.empno%type; 

    v_ename emp.ename%type;  

CURSOR cursor1 IS SELECT empno, ename FROM emp; 

BEGIN 

    OPEN cursor1; 

    LOOP 

        FETCH cursor1 INTO v_empno, v_ename; 

        EXIT WHEN cursor1%ROWCOUNT>5 OR cursor1%NOTFOUND; 

        INSERT INTO mesaje VALUES (v_empno, v_ename); 

    END LOOP; 

CLOSE cursor1; 

END; 



 

 
 

Pasul 3 – afisarea continutului noii tabele MESAJE: 

 

 

 

 

SELECT * FROM mesaje; 



 
 

Scrieti un program in PL/SQL pentru a afisa, cu ajutorul unui cursor, toate informatiile despre 

angajati din tabela EMP.  

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

DECLARE  

    CURSOR c_emp_info IS  

      SELECT empno, ename, sal, deptno  

      FROM   emp;  

    r_emp_info c_emp_info%ROWTYPE;  

BEGIN  

    OPEN c_emp_info;  

    LOOP  

        FETCH c_emp_info INTO r_emp_info;  

        EXIT WHEN c_emp_info%NOTFOUND;  

        dbms_output.Put_line('Employees Information:: '  ||'  ID: ' ||r_emp_info.empno 

||'  Name: ' ||r_emp_info.ename ||' ' ||r_emp_info.sal||' ' ||r_emp_info.deptno);  

    END LOOP;  

    dbms_output.Put_line('Total number of rows : ' ||c_emp_info%rowcount);  

    CLOSE c_emp_info;  

END; 



 

 
 

 

Probleme propuse spre rezolvare 
 

 

 

1. Scrieti un program in PL/SQL pentru a declara o inregistrare cu acelasi tip de date ale 

campurilor din tabela EMP, folosind atributul %TYPE. 

 

2. Scrieti un program in PL/SQL pentru a crea un cursor implicit folosind structura repetitiva 

FOR LOOP. 

 

3. Scrieti un program in PL/SQL pentru a crea un cursor explicit folosind structura repetitiva 

FOR LOOP. 

 

4. Scrieti un program in PL/SQL pentru a mari salariul angajatilor din departamentul 20 

folosind clauza WHERE CURRENT OF intr-un cursor. Marirea salariilor se face astfel: 

Daca salariul este mai mic de 7000 se va mari cu 15%, altfel cu se va mari cu 10%. 

 

5. Scrieti un program in PL/SQL pentru a prelua doar o singura linie din tabela EMP, cu 

ajutorul instructiunii FETCH. 

 

6. Scrieti un program in PL/SQL pentru a prelua mai mult de o singura linie din tabela EMP, 

cu ajutorul instructiunii FETCH. 

 

7. Scrieti un program in PL/SQL pentru a afisa numele si salariul fiecarui angajat din tabela 

EMP al carui salariu este mai mic decat o valoare transmisa ca parametru intr-un cursor. 

 

8. Scrieti un program in PL/SQL pentru a afisa denumirea departamentului, seful 

departamentului, data angajarii si salariul fiecarui angajat din tabela EMP al carui salariu este 

cel mai mare.  

 

9. Scrieti un program in PL/SQL pentru a arata cum se utilizeaza un subquery (subcerere) in 

clauza FROM din interogarea principala a unui curor explicit. 

